
Zusammenfassung 

Im Rahmen klinischer Studien hat die Sicherheit und das Wohlbefinden der Studienteil-
nehmer oberste Priorität. Um dies zu gewährleisten, werden in regelmäßigen Abständen 
die erhobenen Daten von Studienärzten reviewt, um mögliche Risiken zu untersuchen. 
Hierbei helfen grafische Darstellungen der Daten, um die Flut sowie die (mittlerweile) 
große Menge von Daten übersichtlich und einfach darzustellen. Balkendiagramme, darge-
stellt über die Zeit, mit verschiedenen Zeitabschnitten basierend auf dem Studiendesign 
sowie einer Abstufung in Risikokategorien haben sich in der Praxis bewährt und werden 
gern von Studienärzten angefragt. Auch ist diese Methode der Darstellung eine Möglich-
keit, die Wirksamkeit der Studienmedikation individuell zu begutachten. Dieser Beitrag 
präsentiert Ihnen die Erstellung einer solchen Grafik in SAS mit der Graph Template 
Language. 

 
Schlüsselwörter: Patientenprofile, SAS Graph Template Language 

1 Einführung 

In jeder klinischen Studie ist die Sicherheit und das Wohlbefinden der Studienteilneh-
mer von größter Wichtigkeit. Damit dies immer gegeben ist, werden zu bestimmten 
Zeitpunkten Daten erhoben und reviewt. Patientenprofile stellen wichtigste Eckdaten 
des Patienten dar und können einen schnellen Überblick und Verlauf über den Zustand 
des Patienten geben. Aus diesem Grund eignen sich Patientenprofile, die als Sammelsu-
rium aller erhobenen, individuellen Daten erstellt werden, ganz besonders zur Kontrolle 
der Patientensicherheit. Jedoch spielt die Darstellung von solchen Patientenprofilen eine 
ausschlaggebende Rolle. Diese können mit Hilfe von Datenlisten, Tabellen oder Grafi-
ken dargestellt werden. Die grafische Darstellung hat sich in der Praxis besonders be-
währt, da sich unterschiedliche Informationen auf wenigen Seiten übersichtlich über 
Gruppen beziehungsweise Populationen abbilden lassen. 
 
In dieser Publikation wird der Vorteil von grafischen Patientenprofilen und die Erstel-
lung einer Grafik mit SAS vorgestellt. 

2 Darstellung von Patientenprofilen 

Damit Patientenprofile erstellt bzw. betrachtet werden können, wird folgende fiktive 
Studie betrachtet. In dieser fiktiven Studie wurden Daten von neun erkrankten Patienten 
über 40 Wochen erhoben. Die Patienten hatten alle 4 Wochen eine Visite beim Arzt, bei 
der ein Score gemessen wurde. Der Wertebereich des Score liegt zwischen 1 bis 9 und 
wurde anschließend in drei Kategorien eingeteilt: 
 
Kategorie 1: Wert zwischen 1 und 3, 
Kategorie 2: Wert zwischen 4 und 6, 
Kategorie 3: Wert zwischen 7 und 9. 
 
Die Kategorie 3 stellt die schlechteste Kategorie dar.  
 
Die einfachste Methode der Darstellung von Patientenprofilen geht über Datenlisten und 
ist zugleich die gängigste in der Praxis. In Listing 1 ist ein Auszug aus einem fiktiven 
Patientenprofil dargestellt. Zu sehen sind für zwei Patienten die Zentrum- und Patien-
tennummer, das Geschlecht, das Alter, das Stratum, die Visite und die zugehörige Wo-
che, das Datum der Visite, der Score, die Kategorie des Scores, die Einnahme der Studi-
enmedikation, die Einnahme einer Notfallmedikation und ob der Patient aus der Studie 
ausgeschieden ist. 
 
Listing 1: Darstellung eines Patientenprofils in Form eines Datenlistings 

 
Der Vorteil einer Datenliste ist die Darstellung von vielen Informationen auf einer ein-
zelnen Seite – falls die Liste individuell erstellt wird – oder anders gesagt, die Darstel-
lung von vielen Informationen nebeneinander in horizontaler Ausrichtung. Allerdings 
ist ein Trend bezüglich dem Wohlbefinden und der Sicherheit der Arznei (Verbesserung 

oder Verschlechterung der Krankheit) durch Betrachtung der Kategorien nur bei genau-
erem betrachten erkennbar. Ein weiterer Nachteil einer Datenliste ist die Länge des Do-
kuments. Dieses kann schnell über mehrere Seiten gehen und den Reviewer dazu auf-
fordern, die Daten desselben Patienten über mehrere Seiten festzuhalten. In unserem 
Beispiel können maximal zweieinhalb Patienten pro Seite dargestellt werden. Bei neun 
Patienten wären bereits über vier Seiten durchzusehen. Bei größeren Studien mit vielen 
Patienten ist es mit Datenlisten nur bedingt möglich, die Übersicht zu wahren. 
 
Eine andere Möglichkeit der Darstellung von Patientendaten ist die grafische Aufberei-
tung. In Abbildung 1 ist eine Beispielgrafik dargestellt, in der der Krankheitsverlauf je-
des Patienten mit Hilfe von Balkendiagrammen dargestellt ist. Auf der X-Achse ist die 
Zeit abgebildet und bezieht sich auf die Anzahl der Wochen unter Beobachtung. Auf der 
Y-Achse ist die Patienten-ID abgebildet. Zusätzlich zu den Scores, die durch Zahlen in 
den Balken dargestellt sind, werden die Kategorien durch farbliche  Balken hervorgeho-
ben. Grün stellt die Kategorie 1, gelb die Kategorie 2 und rosa die Kategorie 3 dar. Die 
Einnahme der Studienmedikation ist durch kleine Kreise und die Einnahme von Not-
fallmedikation durch Rauten abgebildet. Unerwünschte Ereignisse (engl. Adverse 

Events, [AE]) sind durch dunkel-rote Balken innerhalb der Kategoriebalken dargestellt. 
Der Austritt eines Patienten und der Ausschluss eines Patienten aus der Studie durch 
den Sponsor ist in Grautönen abgebildet. 

 

Abbildung 1: Erste Darstellungsmöglichkeit eines Patientenprofiles. 
 
Abbildung 2 stellt eine Erweiterung von Abbildung 1 dar und enthält zusätzlich noch 
Informationen über das Geschlecht, dem Alter und das zugehörige Stratum. 

Abbildung 2: Zweite Darstellungsmöglichkeit eines Patientenprofiles. 

Der Vorteil der grafischen Darstellung eines Patientenprofiles liegt auf der Hand. Zum 
einen lässt sich durch die farbliche Abbildung der Kategorien 1, 2 und 3 leicht ein 
Überblick über den Verlauf der Krankheit gewinnen.  Zum Anderen sind Trends leich-
ter zu erkennen. Da AEs in einer auffälligen Farbe dargestellt sind, lassen sich diese mit 
einem Blick heraus kristallisieren. Ebenfalls sind fast alle Informationen in den Grafi-
ken abgebildet, wie in der Datenliste vorhanden, sodass im Vergleich zu der Datenliste 
fast kein Informationsverlust vorhanden ist. Durch Hinzufügen von Spalten können 
mehr Informationen dargestellt werden, wie durch das Hinzufügen von Geschlecht, Al-
ter und Stratum in Abbildung 2. Falls noch mehr Informationen gewünscht sind, könnte 
Abbildung 2 durch eine Spalte auf der rechten Seite oder einer weiteren Spalte auf der 
linken Seite erweitert werden. Ein weiterer Vorteil der grafischen Darstellung ist die 
Möglichkeit der Gegenüberstellung von Profilen über mehrere Patienten. Somit lassen 
sich in ein und derselben Abbildung Vergleiche zwischen Patienten darstellen. Diese 
Vergleiche können ganz unterschiedlicher Natur sein und können sich auf Vergleiche 
innerhalb derselben Gruppe beziehen, wie z.B. Patienten, die täglich Aspirin einneh-
men, oder sich auf Vergleiche zwischen verschiedenen Kollektiven, beziehen. 

3 Umsetzung in SAS 

Die Programmierung von Patientenprofilen wurde in dieser Arbeit durch die SAS Graph 
Template Language, kurz GTL, realisiert. In diesem Abschnitt wird das Output Delivery 
System, kurz ODS, und die Erstellung von Grafiken diskutiert. Die Programmierung 
erfolgte in der SAS Version 9.3. 

3.1 SAS Graph Template Language 

Das ODS ist ein System in SAS mit dem die Aufbereitung und die Weiterverarbeitung 
von SAS-Outputs möglich ist. Dabei dient es als Schnittstelle zu anderen Textverarbei-
tungsprogrammen und erlaubt, Texte oder Grafiken in unterschiedlichen Dateiformaten 
zu erstellen und abzulegen. Dazu gehören unter anderem die Formate  rtf, pfd und csv. 
 
Die GTL ist eine Erweiterung des ODS. Dieses bietet die Möglichkeit, „spezielle“ Gra-
fiken zu generieren, welche nicht mit den von SAS vorprogrammierten Prozeduren er-
stellt werden können. Solche Grafiken können zum Beispiel Patientenprofile sein, bei 
denen verschiedene Farben mehrfach und in unterschiedlicher Reihenfolge auftauchen 
sollen. 
 
Die GTL generiert Grafiken anhand von GTL-Vorlagen. Die Vorlagen werden mit der 
SAS-Prozedur PROC TEMPLATE definiert und werden aus Gründen des Umfangs der 
Elemente hier nicht weiter ausgeführt. In diesen definierten Vorlagen werden Layout-
anweisungen, die die Anordnung von Grafikmerkmalen festlegen, Plotanweisungen, die 
bestimmte Plottypen anfordern (wie Histogramme und Streudiagramme), und Text- und 
Legendenanweisungen, die Titel, Fußnoten, Legenden und andere textbasierte Grafik-
elemente festgelegt. Im nächsten Abschnitt werden einige Elemente der GTL, die an-
gewendet wurden, näher beschrieben. 
 
Mit der TEMPLATE-Prozedur wird die Grafik „geformt“, aber nicht gezeichnet. Die 
Erstellung der Grafik erfolgt mit der SAS-Prozedur PROC SGRENDER. Die Prozedur 
benötigt die Plotvariablen (die Daten) sowie eine GTL-Vorlage. Des Weiteren ist die 
Möglichkeit gegeben, den Namen sowie die Bezeichnung der Ausgabeobjekt(e) anzu-
geben. 

3.2 Erstellen der Grafik 

Die Abbildung 2 wurde mit Hilfe von GTL generiert.  Hierbei wurden High-Low- und 
Scatterplots übereinander gelegt (engl. overlay). High-Low-Plots wurden benutzt, um 
die Balken zu zeichnen und farblich zu markieren. Die restlichen Informationen wie 
Score, Medikationseinnahme, Geschlecht, Alter, Stratum wurden jeweils durch ein 
Scatterplot generiert. 
 
Im ersten Schritt  wurde die Grafik beziehungsweise die Rohfassung der Grafik in der 
Prozedur PROC TEMPLATE definiert. Das DEFINE-Statement erlaubt beziehungs-
weise öffnet einen Definitionsblock zur Erstellung einer (Grafik-) Vorlage. Weiterhin 
wurde durch die Option STATGRAPH festgelegt, dass der Output die Struktur und Aus-
sehen einer Grafik haben soll. Danach wurde der Name der Vorlage als „highhlowbar“ 
festgelegt. Alle weiteren Vorgaben an die Grafik werden anschließend innerhalb eines 
BEGINGRAPH/ ENDGRAPH Blocks definiert. 

PROC TEMPLATE; 
  DEFINE STATGRAPH highlowbar; 
    BEGINGRAPH; 
    ... 
    ENDGRAPH; 
  END; 
RUN; 
 

Die Überschrift wurde durch das ENTRYTITLE-Statement definiert. Da die Überschrift 
in unserem Fall über mehrere Zeilen geht, musste das ENTRYTITLE-Statement so oft 
wiederholt werden wie Zeilen erzeugt werden. Durch die Option TEXTATTRS= wurde 
die Schriftgröße der jeweiligen Reihe angepasst. 
 
    ENTRYTITLE HALIGN=CENTER 'Example Graphic 2'; 
    ENTRYTITLE HALIGN=CENTER 'Score by Weeks' / TEXTATTRS=(SIZE=8); 

 
Wie in der Abbildung 2 zu sehen, existieren verschiedene Legendeneinträge, die dem 
Leser das Verständnis der Symbole und Markierungen näher bringen. Innerhalb der 
Vorlagen bietet sich die Möglichkeit, das LEGENDITEM-Statement zu nutzen.  Diese 
erstellt die Definition für ein Legendenelement, das in einer separaten Legende enthal-
ten sein kann. D.h. alle Legeneinträge müssen somit separat definiert werden, um an-
schließend innerhalb einer Legende dargestellt zu werden. Die Option TYPE= definiert, 
welche Art von Symbol in der Legende erscheinen soll und durch die Option NAME= 
wird der Name festgelegt, welcher später beim Aufruf benutzt wird, um die Einträge 
eindeutig zu referenzieren. Unter der Option MARKERATTRS= werden die Farbe, das 
Symbol und die Göße des Symbols festgelegt. Zuletzt wird durch die Option LABEL= 
die Beschreibung des Symbols in der Legende definiert. Ein Beispiel für die Definition 
eines Legendeneintrags ist nachfolgend für ein gefülltes Quadrat der Schriftgröße 7 und 
der Farbe Gainsboro gegeben. Referenziert wird die Definition des Legendeneintrags 
durch den Namen „Drop“, wogegen die Darstellung des Symbols mit „Drop-Out“ in der 
Legende selbst erfolgt. 
 
LEGENDITEM TYPE=MARKER NAME="Drop" / 
   MARKERATTRS=(COLOR=Gainsboro SYMBOL=squareFilled SIZE=7) 
   LABEL="Drop-Out"  
; 
 

Eine Besonderheit der GTL ist der Aufbau auf hierarchischen Anweisungsblöcken, die 
als "Layouts" bezeichnet werden. Der äußerste Anweisungsblock (sprich die erste An-
weisung) bestimmt hierbei 

· Die Gesamtorganisation des Diagramms - unabhängig davon, ob es sich um eine 
Einzel- oder Mehrfachzellendarstellung handelt; 

· Welche Anweisungen im Block erlaubt sind, beispielsweise ist für 3D-Grafiken 
eine andere Umgebung zu wählen als für 2D-Grafiken; 

· Wie die benutzten Statements untereinander agieren. 

Dem Benutzer stehen hierbei verschiedene LAYOUT-Statements zur Verfügung, unter 
anderem OVERLAY, OVERLAY3D, GRIDDED, LATTICE, etc. Zur Erstellung der 
Abbildung 2 wurde ausschließlich das OVERLAY-Layout benutzt. 
 
Mit dem Layout OVERLAY wurde festgelegt, dass mehrere Plots, in der Reihenfolge in 
der sie definiert wurden, auch generiert werden. Dabei teilen sich die Plots die vorher 
festgesetzten Achsen-Grenzen, die über die Optionen YAXISOPTS und XAXISOPTS 
definiert wurden. In der Option LABEL= wird die Beschreibung der Achse festgelegt 
und durch die Option LABELATTRS= die Schriftgröße, die verwendet werden soll. 
Durch die Option GRIDDISPLAY= kann der Benutzer angeben, ob ein Gitter ange-
zeigt werden soll oder nicht. Unter der Option TICKVALUELIST= können die Werte, 
die auf der Achse angegeben werden sollen, definiert werden. Dabei kann durch die Op-
tion TICKVALUEFORMAT= das Format, die Optionen VIEWMIN= und VIEWMAX= 
die Bandbreite und die Option TICKVALUEATTRS= die zu verwendende Schriftgröße  
der Werte definert werden.  Jedoch muss beachtet werden, dass die Optionen 
TICKVALUEATTRS=, TICKVALUEFORMAT, VIEWMIN= und VIEWMAX= in der 
Option LINEAROPTS= definiert werden müssen. Zuletzt wird durch die Option 
DISPLAY= angegeben, was auf der Achse abgebildet werden soll. Ein Beispiel für die 
Definition einer Achse ist nachfolgend für die Y-Achse gegeben. Als Beschreibung der 
Y-Achse wurde der Text „Subject - ID“ in Schriftgröße 8 festgelegt. Um die Patienten-
IDs auf der Y-Achse in chronologischer Reihenfolge darzustellen, wurde ein kleiner 
Trick angewandt. Dazu wurde in den Daten jedem Patienten eine negative Zahl zwi-
schen -1 und -9 zugeordnet und die zugehörige Patienten-ID als Format gespeichert. 
Diese negativen Zahlen konnten dann unter der Option TICKVALUELIST= und das 
Format der Zahlen (formatid.) unter der Option TICKVALUEFORMAT= angegeben 
werden. 
 
Hierbei ist es erforderlich, dass Format vor dem Aufruf der Prozedur SGRENDER zu 
definieren. Dieser wird in unserem Programm dynamisch über PROC SQL erstellt und 
mit Hilfe der Prozedur FORMAT definiert. Sei hier auf die SAS-Hilfe zur Erstellung 
von Makrovariablen mit der SQL-Prozedur verwiesen. 
 
LAYOUT OVERLAY / 
YAXISOPTS=( 
        LABEL="Subject - ID" 
        LABELATTRS=(SIZE=8) 
        GRIDDISPLAY=OFF 
        TICKVALUEATTRS=(SIZE=7pt) 
        LINEAROPTS=(  TICKVALUELIST=(-1 -2 -3 -4 -5 -6 -7 -8 -9 ) 
                      TICKVALUEFORMAT=formatid. 
                      VIEWMIN=-9 VIEWMAX=-1 ) 
        DISPLAY=(LABEL LINE TICKS TICKVALUES) 
        ) 
XASISOPTS=(...) 
; 

Die X-Achse wurde nach dem gleichen Prinzip der Y-Achse erstellt. Nachdem die 
Achsenabschnitte definert wurden, konnten für jede Woche vertikale Referenzlinien 
eingezeichnet werden. Dazu musste innerhalb des REFERENCELINE-Statements für 
jede Woche, an der eine Referenzlinie gezeichnet werden sollte, die entsprechende Wo-
che aufgeführt werden. Zur Vereinfachung des Programmcodes wurde eine Macro-
Schleife verwendet, um die Anzahl der verfügbaren Wochen automatisiert in die Defini-
tion der Vorlage aufzunehmen. 
     
%DO i=1 %TO 40; 
  REFERENCELINE X=&i. / LINEATTRS=(COLOR=GWH); 
%END; 
 

Im Anschluss wurden die Balkendiagramme erstellt. Damit die Reihenfolge der Farben 
in den Balken unabhängig bleibt und auch mehrere Balken dieselbe Farbe haben konn-
ten, musste für jeden Balken ein eigener Plot erstellt werden. Dies wurde mit Hilfe des 
HIGHLOWPLOT-Plottyps umgesetzt. Durch die Option Y/X= werden die Werte, 
die auf der Achse dargestellt werden sollen, definiert. Bei der Verwendung von X= wird 
angegeben, dass die Werte auf der X-Achse abgebildet werden und bei Y=, dass die 
Werte auf der Y-Achse abgebildet werden sollen. Unter den Optionen HIGH= und 
LOW= werden die Grenzen der Balken angegeben. Weiterhin können die Daten durch 
die Option GROUP= gruppiert werden. Ob in dem High-Low-Plot Balken oder Linien 
dargestellt werden sollen, kann durch Option TYPE= angegeben werden. Bei der Wahl 
von Balken kann unter der Option OUTLINEATTRS= die Eigensachaften der Umran-
dung der Balken definiert werden. Die Struktur (z.B. gestrichelte oder durchgezogene 
Linie) und die Farbe der Umrandung werden in den Optionen PATTERN= und 
COLOR= angegeben.  Zum Schluss kann die  Füllfarbe des Balken in der Option 
FILLATTRS= durch die Option COLOR= definiert werden. Da für die Generierung 
der Abbildung 2 für jeden Balken ein eigener High-Low-Plot benötigt wurde, wurde bei 
der Programmierung, zur Verkürzung des Programmcodes, eine weitere Makro-Schleife 
verwendet. Die verschiedenen Variablen im Datenasatz, welche die Information der un-
teren und oberen Grenzen der Balken beinhalten, bekamen bei der Programmierung als 
Namen einen Prefix („high“ oder „low“) und eine Zahl zugewiesen, beispielsweise 
high_1. Dadurch konnten die Variablen in der Makro-Schleife durch die Verwendung 
des Schleifenindex leicht aufgerufen werden.  Die Farbe der Balken und der Umran-
dungen wurden als String in einer Makro-Variable gespeichert und konnten in der Mak-
ro-Schleife durch die Verwendung des Schleifenindex und der Funktion %SCAN() auf-
gerufen werden. 
%DO ii=1 %TO 10;   
  HIGHLOWPLOT Y=id HIGH=high&ii. LOW=low&ii. /  
    GROUP=id 
    OUTLINEATTRS=(PATTERN=SOLID COLOR=%SCAN(&_colors.,&ii.,#)) 
    FILLATTRS=(COLOR=%SCAN(&_colors.,&ii.,#)) 
    TYPE=BAR 
  ; 
%END; 

Informationen wie die Einnahme der Studien- und Notfallmediakation wurden durch die 
Verwendung von Scatterplots eingezeichnet. Durch die Optionen X= und Y= können die 
Koordinaten an denen die Scatter abgebildet werden sollen, übergeben werden. Durch 
die Option MARKERATTRS= können die Symbolgröße (SIZE=), Farbe (COLOR=) und 
das Symbol selbst (SYMBOL=) definiert werden. Im nachfolgendem Beispiel ist die De-
finition für die Einnahme der Studienmedikation, in Form eines schwarzen Kreises, in 
der Symbolgröße 6 gegeben. 
 
SCATTERPLOT Y=medicationid X=medicationWeek /  
      MARKERATTRS=(SIZE=6 SYMBOL=Circle COLOR=black); 

 

Im Gegensatz zur  Einnahme der Studien- und Notfallmedikation sollten die Informati-
onen über Geschlecht, Alter, Stratum nicht als Symbol, sondern als die Werte selbst 
dargestellt werden. Dazu kann in der Prozedur SCATTERPLOT die Option 
MARKERCHARACTER= verwendet werden. Dadurch werden die Werte der angegebe-
nen Variablen dargestellt. 
 
SCATTERPLOT Y=id X=scoreWeek / 
      MARKERCHARACTER=score MARKERATTRS=(SIZE=19px); 
 

Zum Schluss wurde die Legende durch das DISCRETELEGEND-Statement aufgerufen. 
Durch das Auflisten der Namen der vorher definierten Legendenelemente wird festge-
legt, welche Elemente in der Legende auftauchen. Dabei werden die Legendenelemente, 
in der Reihenfolge, in der sie aufgelistet sind, dargestellt. Mit der Option ACROSS= 
wird angegeben, in wie vielen Spalten die Legendenelemente abgebildet werden sollen. 
Der Titel der Legende  wird durch die Option TITLE= und die Schriftgröße des Titels 
durch die Option TITLEATTRS=(SiZE=) definiert. Weiterhin wird die Position der 
Legende in der Grafik durch die Option LOCATION= bestimmt.  Zuletzt wird in der 
Option OPAQUE=TRUE/FALSE angegeben, ob der Hintergrund der Legende opaque 
oder transparent sein soll. 
       DISCRETELEGEND "Cat1" "Adverse" "Medication" "Score" "Cat2" 
                      "Drop" "Rescue" "Exclusion" "Cat3" "Excluded"/ 
          TITLE="Colors/ Symbols:" 
          TITLEATTRS=(SIZE=8 ) 
          VALUEATTRS=(SIZE=7) 
          OPAQUE=TRUE 
          ACROSS=4 
          LOCATION=OUTSIDE 
        ; 

Nachdem die Vorlage zur Erstellung der gewünschten Grafik definiert wurde, konnte 
die Grafik mit Hilfe der Prozedur PROC SGRENDER generiert werden. Um der Grafik 
bestimmte Rahmenbedingungen zuzuweisen, wurde die Prozedur in eine ODS 
LISTING Umgebung mit der ODS GRAPHICS Option eingebettet. In der ODS 
LISTING Umgebung konnte durch die Option IMAGE_DPI= die Auflösung der Gra-
fik erhöht werden.  Durch die ODS GRAPHICS Option konnte der Name 

(IMAGENAME=), die Weite (WIDTH=), die Höhe (HEIGHT=)  und das Grafikformat 
(IMAGEFMT=) der Grafik angegeben werden. Weiterhin wurden durch die Option 
RESET alle nicht definierten Optionen auf ihren Default Einstellung gesetzt. Zuletzt 
wurde durch die Verwendung der Option NOBORDER definiert, dass um die Grafik 

kein äußerer Rand abgebildet werden soll. 

ODS LISTING IMAGE_DPI=300; 
ODS GRAPHICS ON / RESET NOBORDER WIDTH=750px HEIGHT=410px 
    IMAGENAME="Example Graphic 2" IMAGEFMT=png; 

  PROC SGRENDER DATA=highlowbar TEMPLATE=highlowbar; 
  RUN; 

ODS GRAPHICS OFF; 
ODS LISTING CLOSE; 

4 Zusammenfassung 

Bei jeder klinischen Studie hat die Sicherheit und das Wohlbefinden der Studienteil-
nehmer oberste Priorität. Patientenprofile eignen sich besonders, um die gesammelten 
Daten im Verlauf der Studie abzubilden und zu begutachten. Die einfachste Methode ist 
die Darstellung von Patientenprofile in einer Datenliste. Dies hat den Vorteil, dass viele 
Informationen nebeneinander (in Spalten) dargestellt werden können. Jedoch hat eine 
Datenliste zwei entscheidene Nachteile. Erstens können Trends (Veschlechterung oder 
Verbesserung der Krankheit) schlecht erkannt werden.  Zweitens können Datenlisten 
schnell über viele Seiten gehen und dadurch unübersichtlich werden. 

Die grafische Abbildung von Patientenprofilen ist neben der Datenliste eine weitere 
Darstellungsmöglichkeit. Die Vorteile der grafischen Darstellung lassen sich folgern-
dermaßen auflisten: 

- Farbliche Abbildung von Kategorien ermöglichen einen leichten Überblick über 
den Verlauf der Krankheit 

- Trends der Krankheit sind gut zu erkennen 
- AEs in auffälligen Farben dargestellt, stechen ins Auge 
- Fast kein Informationsverlust im Vergleich zur Datenliste 
- Patientenspezifische Informationen können durch Spalten abgebildet werden 
- Möglichkeit der Gegenüberstellung von Profilen über mehrere Patienten 

Grafische Patientenprofile können mit GTL in SAS generiert werden. Dazu werden in 
einem STATGRAPH-Template Informationen über Layout, und der Achsen definiert 
und durch die Option OVERLAY meherere Plots übereinander gelegt. Die Grafik wird 
durch das Aufrufen der Prozedur SGRENDER erstellt und durch das Hinzufügen einer 
ODS LISTING Umgebung mit dem ODS GRAPHICS-Statement mit weiteren Rah-
menbedingungen definiert werden. 

